program mynormal5
	version 9.1
	if replay() {
		if (`"`e(cmd)'"' != "mynormal5") error 301
		Replay `0'
	}
	else	Estimate `0'
end

program Estimate, eclass sortpreserve
	syntax varlist [if] [in]		///
		[fweight pweight] [,		/// <- NEW
		noLOg				/// -ml model- options
		noCONStant			///
		CLuster(varname)		///
		HETero(varlist)			///
		Level(cilevel)			/// -Replay- option
		*				/// -mlopts- options
	]

	// check syntax
	mlopts mlopts, `options'
	gettoken lhs rhs : varlist
	if "`cluster'" != "" {
		local clopt cluster(`cluster')
	}
	if "`weight'" != "" {				// NEW block
		local wgt "[`weight'`exp']"
	}
	// mark the estimation sample
	marksample touse
	markout `touse' `hetero'			// <- CHANGED
	markout `touse' `cluster', strok

	// fit the full model
	ml model d2 mynormal_d2				///
		(mu: `lhs' = `rhs', `constant')		///
		(lnsigma: `hetero')			///
		`wgt' if `touse',			/// <- CHANGED
		`log'					///
		`mlopts'				///
		`clopt'					///
		missing					///
		maximize

	ereturn local cmd mynormal5

	Replay , level(`level')
end

program Replay
	syntax [, Level(cilevel) ]
	ml display , level(`level')
end
